

# ALGORITMOS E LÓGICA DE PROGRAMAÇÃO

## Estruturas de Controle

Profa. Dra. Jaqueline Brigladori Pugliesi

## Estruturas de Controle

- Estrutura Sequencial
- Estrutura de Seleção
- Estrutura de Repetição



## Estrutura Sequencial



## Estrutura Sequencial

- O Fluxo de Controle segue a mesma sequência linear da nossa escrita, ou seja:
  - De cima para baixo;
  - Da esquerda para direita
- Cada ação é seguida de um ;
  - Objetiva separar uma ação da outra
  - Indica que a próxima ação da sequência deve ser executada

## Estrutura Sequencial

```
programa
{
    funcao inicio()
    {
        // Declaração de variáveis
        real N1, N2, N3, N4, MA
        // Entrada de dados
        escreva("Digite quatro valores: \n")
        leia(N1, N2, N3, N4)
        // Processamento
        MA = (N1 + N2 + N3 + N4) / 4
        // Saída de dados
        escreva("A média é ", MA)
    }
}
```

## Estrutura de Seleção

## Estruturas de Seleção

- São aquelas que permitem alterar o Fluxo de Execução, de forma a selecionar qual parte deve ser executada
- Essa “decisão” de execução é tomada a partir de uma condição, que pode resultar apenas em Verdade ou Falsidade
- Uma condição é representada por expressões relacionais ou lógicas
- As estruturas de seleção podem ser classificadas em simples, compostas ou encadeadas.

## Seleção Simples

```
se (condicao) {  
    <sequência-de-comandos-1>  
}
```

## Seleção Simples

### Média Aritmética com Aprovação

```
programa
{
    funcao inicio()
    {
        real N1, N2, N3, N4, MA
        escreva("Entre com as quatro notas: \n")
        leia(N1, N2, N3, N4)
        MA = (N1 + N2 + N3 + N4) / 4
        escreva("A média é ", MA, "\n")
        se (MA >= 7)
        {
            escreva("Aluno Aprovado!\n")
        }
    }
}
```

## Seleção Composta

```
se (condicao) {
    <sequência-de-comandos-1>
}
senao {
    <sequência-de-comandos-2>
}
```

## Seleção Composta

Média Aritmética com aprovação e reprovação

```
programa
{
    funcao inicio()
    {
        real N1, N2, N3, N4, MA
        escreva("Entre com as quatro notas: \n")
        leia(N1, N2, N3, N4)
        MA = (N1 + N2 + N3 + N4) / 4
        escreva("A média é ", MA, "\n")
        se (MA >= 7)
        {
            escreva("Aluno Aprovado!\n")
            escreva("Parabéns!\n")
        }
        senao
        {
            escreva("Aluno Reprovado!\n")
            escreva("Estude mais!\n")
        }
    }
}
```

## Exercícios

1. Fazer um algoritmo que leia um número e imprima se o número é par ou ímpar.
2. Ler um número inteiro e imprimir se ele é par e divisível por três.
3. Escreva um algoritmo que leia um número e imprima se o número é positivo ou negativo.
4. Escreva um algoritmo que leia dois valores inteiros e diferentes, e mostre-os em ordem decrescente.
5. Crie um programa que leia uma palavra e verifique se a palavra digitada é igual a Algoritmo. Mostre uma mensagem dizendo ALG, caso tenha sido digitado Algoritmo.

## Seleção Encadeada

- Ocorre quando uma seleção tem como ação uma outra seleção
- Uma seleção encadeada pode ser:
  - Heterogênea: quando não é possível identificar padrão de comportamento
  - Homogênea: quando é possível identificar padrão de comportamento
    - se – então – se: quando depois de cada então ocorre outro se
    - se – senão – se: quando depois de cada senão ocorre outro se

## RH de uma empresa

- Um RH está recrutando candidatos para entrevistas para uma vaga de emprego.
- A vaga pode ser ocupada por pessoas do sexo Feminino ou do sexo Masculino, porém, para pessoas do sexo Feminino, só serão aceitas pessoas maiores de 21 anos.
- Para as pessoas do sexo masculino, serão aceitas pessoas maiores de 18 anos. Os demais candidatos não serão chamados para entrevistas.
- Os homens deverão ter carteira de habilitação tipo AB.

# Seleção Encadeada Homogênea

```
se - então - se
se <Cond1> então
  se <Cond2> então
    se <Cond3> então
      se <Cond4> então W;
    fimse;
  fimse;
fimse;
```

Cond1	Cond2	Cond3	Cond4	Ação
V	V	V	V	W

É equivalente a:

```
se <Cond1> e <Cond2> e <Cond3> e <Cond4> então W;
fimse;
```

# Seleção Encadeada Homogênea

```
se X=V1 então
  C1;
fimse;
se X=V2 então
  C2;
fimse;
se X=V3 então
  C3;
fimse;
se X=V4 então
  C4;
fimse;
```

X=V1	X=V2	X=V3	X=V4	Ação
V	F	F	F	C1
F	V	F	F	C2
F	F	V	F	C3
F	F	F	V	C4

```
se X=V1
  então C1;
  senão se X=V2
    então C2;
    senão se X=V3
      então C3;
      senão se X=V4
        então C4;
        fimse;
    fimse;
  fimse;
```

se – senão – se

X=V1	X=V2	X=V3	X=V4	Ação
V	-	-	-	C1
F	V	-	-	C2
F	F	V	-	C3
F	F	F	V	C4



## Exemplo

Escreva um algoritmo que leia um número e imprima se o número é positivo, negativo ou zero.

```
programa
{
    funcao inicio()
    {
        inteiro num
        escreva("Digite um número: ")
        leia(num)
        se (num > 0)
        {
            escreva("Positivo")
        }
        senao
        {
            se (num < 0)
            {
                escreva("Negativo")
            }
            senao
            {
                escreva("Zero")
            }
        }
    }
}
```

## Exemplo

Fazer um algoritmo que leia 3 valores inteiros e verifique se eles podem formar um triângulo. Se for possível formar um triângulo, escreva uma mensagem informando se é um triângulo equilátero, isósceles ou escaleno.

Observações:

- O comprimento de um lado do triângulo é sempre menor do que a soma dos outros dois.
- Equilátero: todos lados iguais
- Isósceles: dois lados iguais
- Escaleno: todos os lados diferentes

```
programa
{
    funcao inicio()
    {
        inteiro A, B, C
        escreva("Digite três valores: \n")
        leia(A, B, C)
        se (A < B+C e B < A+C e C < A+B)
        {
            se (A == B e B == C)
            {
                escreva("Triângulo Equilátero")
            }
            senao
            {
                se (A == B ou B == C ou A == C)
                {
                    escreva ("Triângulo Isósceles")
                }
                senao
                {
                    escreva ("Triângulo Escaleno")
                }
            }
        }
        senao
        {
            escreva ("Estes valores não formam um triângulo")
        }
    }
}
```

## Exercícios

6. Tendo como dados de entrada a altura e o sexo de uma pessoa, construa um algoritmo que calcule o seu peso ideal, utilizando as seguintes fórmulas:
- para homens:  $(72.7 * h) - 58$ ;
  - para mulheres:  $(62.1 * h) - 44.7$

## Exercícios

7. Elabore um algoritmo que leia o valor de dois números inteiros e a operação aritmética desejada. Calcule então a resposta adequada. Utilize os símbolos da tabela a seguir para ler qual a operação aritmética foi escolhida:

Símbolo	Operação aritmética
+	Adição
-	Subtração
*	Multiplicação
/	Divisão

## Exercícios

8. Escreva um algoritmo que leia três valores inteiros e diferentes e mostre-os em ordem decrescente. Utilize para tal uma seleção encadeada.

## Exercícios

9. Elabore um algoritmo que calcule o que deve ser pago por um produto, considerando o preço normal de etiqueta e a escolha da condição de pagamento. Utilize os códigos da tabela a seguir para ler qual a condição de pagamento escolhida e efetuar o cálculo adequado.

COD	Condição de pagamento escolhida
1	A vista em dinheiro ou cheque, 10% de desconto
2	A vista no cartão de crédito, 5% de desconto
3	Em duas vezes, preço normal de etiqueta sem juros
4	Em três vezes, preço normal de etiqueta mais juros de 10%

## Exercícios

10. A jornada de trabalho semanal de um funcionário é de 40 horas. O funcionário que trabalhar mais de 40 horas receberá hora extra, cujo cálculo é o valor da hora regular com um acréscimo de 50%. Escreva um algoritmo que leia o número de horas trabalhadas em um mês, o salário por hora e escreva o salário total do funcionário, que deverá ser acrescido das horas extras, caso tenham sido trabalhadas (considere que o mês possua 4 semanas exatas).

## Exercícios

11. Faça um algoritmo para ler: quantidade atual em estoque, quantidade máxima em estoque e quantidade mínima em estoque de um produto. Calcular e escrever a quantidade média ( $\text{quantidade média} = (\text{quantidade máxima} + \text{quantidade mínima}) / 2$ ). Se a quantidade em estoque for maior ou igual a quantidade média escrever a mensagem 'Não efetuar compra', senão escrever a mensagem 'Efetuar compra'.

## Exercícios

12. Um posto está vendendo combustíveis com a seguinte tabela de descontos:

Álcool	até 20 litros, desconto de 3% por litro
	acima de 20 litros, desconto de 5% por litro
Gasolina	até 20 litros, desconto de 4% por litro
	acima de 20 litros, desconto de 6% por litro

Escreva um algoritmo que leia o número de litros vendidos e o tipo de combustível (codificado da seguinte forma: A-álcool, G-gasolina), calcule e imprima o valor a ser pago pelo cliente sabendo-se que o preço do litro da gasolina é R\$ 3,30 e o preço do litro do álcool é R\$ 2,10.

# FIM